

**pergram** — Periodogram

[Description](#)  
[Options](#)  
[Also see](#)

[Quick start](#)  
[Remarks and examples](#)

[Menu](#)  
[Methods and formulas](#)

[Syntax](#)  
[References](#)

## Description

`pergram` plots the log-standardized periodogram for a dense time series.

## Quick start

Plot periodogram of `y` using `tsset` data

```
pergram y
```

As above, but generate variable `newvar` to hold raw periodogram values

```
pergram y, generate(newvar)
```

As above, but suppress display of periodogram graph

```
pergram y, generate(newvar) nograph
```

## Menu

Statistics > Time series > Graphs > Periodogram

## Syntax

```
pergram varname [if] [in] [, options]
```

| <i>options</i>                          | Description  |
|---|--|
| Main                                    |  |
| <code>generate(<i>newvar</i>)</code>    | generate <i>newvar</i> to contain the raw periodogram values                             |
| Plot                                    |  |
| <code>cline_options</code>              | affect rendition of the plotted points connected by lines                                |
| <code>marker_options</code>             | change look of markers (color, size, etc.)   |
| <code>marker_label_options</code>       | add marker labels; change look or position   |
| Add plots                               |  |
| <code>addplot(<i>plot</i>)</code>       | add other plots to the generated graph   |
| Y axis, X axis, Titles, Legend, Overall |  |
| <code>twoway_options</code>             | any options other than <code>by()</code> documented in [G-3] <code>twoway_options</code> |
| <code>nograph</code>                    | suppress the graph   |

You must `tsset` your data before using `pergram`; see [TS] `tsset`. Also, the time series must be dense (nonmissing with no gaps in the time variable) in the specified sample.

*varname* may contain time-series operators; see [U] 11.4.4 Time-series varlists.

`nograph` does not appear in the dialog box.

## Options

### Main

`generate(newvar)` specifies a new variable to contain the raw periodogram values. The generated graph log-transforms and scales the values by the sample variance and then truncates them to the  $[-6, 6]$  interval before graphing them.

### Plot

`cline_options` affect the rendition of the plotted points connected by lines; see [G-3] `cline_options`.

`marker_options` specify the look of markers. This look includes the marker symbol, the marker size, and its color and outline; see [G-3] `marker_options`.

`marker_label_options` specify if and how the markers are to be labeled; see [G-3] `marker_label_options`.

### Add plots

`addplot(plot)` adds specified plots to the generated graph; see [G-3] `addplot_option`.

### Y axis, X axis, Titles, Legend, Overall

`twoway_options` are any of the options documented in [G-3] `twoway_options`, excluding `by()`. These include options for titling the graph (see [G-3] `title_options`) and for saving the graph to disk (see [G-3] `saving_option`).

The following option is available with `pergram` but is not shown in the dialog box: `nograph` prevents `pergram` from constructing a graph.

## Remarks and examples

[stata.com](http://www.stata.com)


A good discussion of the periodogram is provided in [Chatfield \(2004\)](#), [Hamilton \(1994\)](#), and [Newton \(1988\)](#). Chatfield is also a good introductory reference for time-series analysis. Another classic reference is [Box, Jenkins, and Reinsel \(2008\)](#). `pergram` produces a scatterplot in which the points of the scatterplot are connected. The points themselves represent the log-standardized periodogram, and the connections between points represent the (continuous) log-standardized sample spectral density.

In the following examples, we present the periodograms with an interpretation of the main features of the plots.


### ► Example 1

We have time-series data consisting of 144 observations on the monthly number of international airline passengers (in thousands) between 1949 and 1960 ([Box, Jenkins, and Reinsel 2008](#), Series G). We can graph the raw series and the log periodogram for these data by typing

```
. use http://www.stata-press.com/data/r15/air2
(TIMESLAB: Airline passengers)
. scatter air time, m(o) c(1)
```



```
. pergram air
```




The periodogram highlights the annual cycle together with the harmonics. Notice the peak at a frequency of about 0.08 cycles per month (cpm). The period is the reciprocal of frequency, and the reciprocal of 0.08 cpm is approximately 12 months per cycle. The similarity in shape of each group of 12 observations reveals the annual cycle. The magnitude of the cycle is increasing, resulting in the peaks in the periodogram at the harmonics of the principal annual cycle.

◀


## ▶ Example 2

This example uses 215 observations on the annual number of sunspots from 1749 to 1963 (Box and Jenkins 1976, Series E). The graph of the raw series and the log periodogram for these data are given as

```
. use http://www.stata-press.com/data/r15/sunspot
(TIMESLAB: Wolfer sunspot data)
. scatter spot time, m(o) c(1)
```



```
. pergram spot
```




The periodogram peaks at a frequency of slightly less than 0.10 cycles per year, indicating a 10- to 12-year cycle in sunspot activity.

◀


### ▶ Example 3

Here we examine the number of trapped Canadian lynx from 1821 through 1934 (Newton 1988, 587). The raw series and the log periodogram are given as

```
. use http://www.stata-press.com/data/r15/lynx2
(TIMESLAB: Canadian lynx)
. scatter lynx time, m(o) c(1)
```



```
. pergram lynx
```




The periodogram indicates that there is a cycle with a duration of about 10 years for these data but that it is otherwise random.

◀


### ▶ Example 4

To more clearly highlight what the periodogram depicts, we present the result of analyzing a time series of the sum of four sinusoids (of different periods). The periodogram should be able to decompose the time series into four different sinusoids whose periods may be determined from the plot.

```
. use http://www.stata-press.com/data/r15/cos4
(TIMESLAB: Sum of 4 Cosines)
. scatter sumfc time, m(o) c(1)
```



```
. pergram sumfc, gen(ordinate)
```



The periodogram clearly shows the four contributions to the original time series. From the plot, we can see that the periods of the summands were 3, 6, 12, and 36, although you can confirm this by using

```
. generate double omega = (_n-1)/144
. generate double period = 1/omega
(1 missing value generated)
. list period omega if ordinate > 1e-5 & omega <= .5
```

|     | period | omega     |
|-----|--------|-----------|
| 5.  | 36     | .02777778 |
| 13. | 12     | .08333333 |
| 25. | 6      | .16666667 |
| 49. | 3      | .33333333 |

4

## Methods and formulas

We use the notation of [Newton \(1988\)](#) in the following discussion.

A time series of interest is decomposed into a unique set of sinusoids of various frequencies and amplitudes.

A plot of the sinusoidal amplitudes (ordinates) versus the frequencies for the sinusoidal decomposition of a time series gives us the spectral density of the time series. If we calculate the sinusoidal amplitudes for a discrete set of “natural” frequencies ( $1/n, 2/n, \dots, q/n$ ), we obtain the periodogram.

Let  $x(1), \dots, x(n)$  be a time series, and let  $\omega_k = (k-1)/n$  denote the natural frequencies for  $k = 1, \dots, (n/2) + 1$ . Define

$$C_k^2 = \frac{1}{n^2} \left| \sum_{t=1}^n x(t) e^{2\pi i(t-1)\omega_k} \right|^2$$

A plot of  $nC_k^2$  versus  $\omega_k$  is then called the periodogram.

The sample spectral density is defined for a continuous frequency  $\omega$  as

$$\hat{f}(\omega) = \begin{cases} \frac{1}{n} \left| \sum_{t=1}^n x(t) e^{2\pi i(t-1)\omega} \right|^2 & \text{if } \omega \in [0, .5] \\ \hat{f}(1 - \omega) & \text{if } \omega \in [.5, 1] \end{cases}$$

The periodogram (and sample spectral density) is symmetric about  $\omega = 0.5$ . Further standardize the periodogram such that

$$\frac{1}{n} \sum_{k=2}^n \frac{nC_k^2}{\hat{\sigma}^2} = 1$$

where  $\hat{\sigma}^2$  is the sample variance of the time series so that the average value of the ordinate is one.

Once the amplitudes are standardized, we may then take the natural log of the values and produce the log periodogram. In doing so, we truncate the graph at  $\pm 6$ . We drop the word “log” and simply refer to the “log periodogram” as the “periodogram” in text.

## References

- Box, G. E. P., and G. M. Jenkins. 1976. *Time Series Analysis: Forecasting and Control*. Oakland, CA: Holden-Day.
- Box, G. E. P., G. M. Jenkins, and G. C. Reinsel. 2008. *Time Series Analysis: Forecasting and Control*. 4th ed. Hoboken, NJ: Wiley.
- Chatfield, C. 2004. *The Analysis of Time Series: An Introduction*. 6th ed. Boca Raton, FL: Chapman & Hall/CRC.
- Hamilton, J. D. 1994. *Time Series Analysis*. Princeton, NJ: Princeton University Press.
- Newton, H. J. 1988. *TIMESLAB: A Time Series Analysis Laboratory*. Belmont, CA: Wadsworth.

## Also see

- [TS] **tsset** — Declare data to be time-series data
- [TS] **corrgram** — Tabulate and graph autocorrelations
- [TS] **cumsp** — Cumulative spectral distribution
- [TS] **wntestb** — Bartlett’s periodogram-based test for white noise